GEOFORSCHUNGSZENTRUM POTSDAM
STIFTUNG DES ÖFFENTLICHEN RECHTS

Zdeněk Martinec
Detlef Wolf

Explicit form of the propagator matrix for a multi-layered, incompressible viscoelastic sphere

Scientific Technical Report STR98/08

Impressum

GeoForschungsZentrum Potsdam
Telegrafenberg
D-14473 Potsdam

e-mail: postmaster@gfz-potsdam.de
www: http://www.gfz-potsdam.de

Gedruckt in Potsdam
Mai 1998

A 1.7.1

Zdeněk Martinec, Detlef Wolf

Explicit form of the propagator matrix for a multi-layered, incompressible viscoelastic sphere

08. SEP. 1998

GeoForschungsZentrum Potsdam
STR 98/08
- Bibliothek -
Telegrafenberg, 14473 Potsdam

Scientific Technical Report STR98/08

Explicit form of the propagator matrix for a multi-layered, incompressible viscoelastic sphere

Zdeněk Martinec [1] and Detlef Wolf

GeoForschungsZentrum Potsdam, Division 1: Kinematics and Dynamics of the Earth, Telegrafenberg A17, D-14473, Potsdam, Germany

Abstract

The layer propagator matrix for the response of an incompressible, layered linear-viscoelastic sphere to an external load is derived in analytical form. Its explicit dependence on the Laplace-transform variable s allows us to determine the amplitudes of viscous modes analytically. This improves the numerical accuracy when computing the elastic and viscous amplitudes of the response.

1. Introduction

The theory of the glacial-isostatic adjustment process belongs to the classical problems of solid earth geophysics and has been developed over several decades. To make the forward and inverse problems tractable, the Earth is usually modelled as viscoelastic, self-gravitating, and spherically symmetric. Particular attention has been paid to an incompressible Maxwell viscoelastic solid, since this is the simplest rheology that describes the short-time (elastic) and long-time (inviscid) limits of the Earth's response correctly.

The literature dealing with the theory of viscoelastic gravitational relaxation of an incompressible earth model is quite extensive. A short overview over the studies on incompressible earth models completed during the last two decades could be as follows. Wu & Peltier (1982) derived the analytical formulae for the response of a homogeneous, Maxwell sphere. Sabadini et al. (1982) extended their study and introduced semi-analytical solutions for the relaxation of two-layer and three-layer models of the Earth. Spada et al. (1992) inverted the fundamental matrix associated with field equations by a symbolic manipulation software and gave an analytical expression for the inverse of the fundamental matrix. Wolf (1984) and Amelung & Wolf (1994) presented closed-form solutions for the viscoelastic relaxation of an earth model composed of a viscoelastic mantle and an inviscid core or an elastic lithosphere. Wu (1990) analysed gravitational-viscoelastic perturbations of two-layer spheres with arbitrary contrasts of density, shear modulus and viscosity across the interfaces. Recently, Wu & Ni (1996) have provided some analytical solutions for the viscoelastic relaxation of two-layer, non-self-gravitating spherical earth models. The formulae derived by a symbolic manipulation software and introduced in the appendix show the complexity of the analytical solutions and/or the still imperfect algebraic manipulation capability of today's symbolic manipulation software.

Despite these extensive theoretical investigations, the layer propagator matrix for a homogeneous, incompressible viscoelastic spherical layer has not been published yet. Wu & Ni (1996)

[1] On leave from Department of Geophysics, Faculty of Mathematics and Physics, Charles University, V Holešovičkách 2, 180 00 Prague 8, Czech Republic

suggested a probable explanation of this: 'Even with the help of symbolic manipulation software, general expressions for the analytical solutions of a two-layer self-gravitating earth are generally too long to be analysed or reported.' In this paper, we partly refute this argument and express analytically the propagator matrix for a homogeneous self-gravitating incompressible viscoelastic spherical layer. The layer propagator matrix is not very complicated in form, so that can be used for a further analysis of the elastic and viscous amplitudes.

2. Matrix of fundamental solutions

The response of a self-gravitating, viscoelastic earth model to the gravitational interaction with a surface mass load is governed by the equation of momentum conservation, the equation of mass conservation, Poisson's equation and the constitutive equation. In the Laplace-transform domain s, these field equations for a spherically symmetric body can be written in terms of a set of six simultaneous first-order ordinary differential equations in the form (e.g., Peltier, 1974; Cathles, 1975)

$$\frac{d\boldsymbol{Y}}{dr} = \boldsymbol{A}\boldsymbol{Y} \; , \tag{1}$$

where the elements of the vector $\boldsymbol{Y}(r,s) = (U_n, V_n, T_{rn}, T_{\vartheta n}, \Phi_n, Q_n)^T$ are the spherical harmonic coefficients of radial and tangential displacements, radial and tangential stresses, and perturbations in gravitational potential and potential gradient, respectively. Here r denotes the geocentric radial distance, ϑ the co-latitude, n the angular degree, and $\boldsymbol{A}$ is a 6×6 matrix whose elements depend on r and s. On the assumption of incompressibility, the matrix $\boldsymbol{A}$ reads

$$\boldsymbol{A} = \begin{pmatrix} -\frac{2}{r} & \frac{n(n+1)}{r} & 0 & 0 & 0 & 0 \\ -\frac{1}{r} & \frac{1}{r} & 0 & \frac{1}{\mu} & 0 & 0 \\ \frac{4(3\mu - r\varrho g)}{r^2} & \frac{n(n+1)(-6\mu + r\varrho g)}{r^2} & 0 & \frac{n(n+1)}{r} & -\frac{(n+1)\varrho}{r} & \varrho \\ \frac{-6\mu + r\varrho g}{r^2} & \frac{2\mu(2n^2+2n-1)}{r^2} & -\frac{1}{r} & -\frac{3}{r} & \frac{\varrho}{r} & 0 \\ -4\pi G\varrho & 0 & 0 & 0 & -\frac{n+1}{r} & 1 \\ -\frac{4\pi G\varrho(n+1)}{r} & \frac{4\pi G\varrho n(n+1)}{r} & 0 & 0 & 0 & \frac{n-1}{r} \end{pmatrix} \tag{2}$$

with $\varrho(r)$ the unperturbed density, $g(r)$ the unperturbed gravitational acceleration, G Newton's gravitational constant, and $\mu(r,s)$ the s-dependent shear modulus. For the special case of Maxwell viscoelasticity, it has the form (e.g., Cathles, 1975)

$$\mu(s) = \frac{\mu_e s}{s + \mu_e/\nu} \; , \tag{3}$$

where $\mu_e(r)$ is the elastic shear modulus and $\nu(r)$ is the viscosity.

The general solution to eqn.(1) can be expressed as a sum of six linearly independent fundamental solution vectors:

$$\boldsymbol{Y}(r,s) = \boldsymbol{M}(r,s)\boldsymbol{C}(s) \; , \tag{4}$$

where $\boldsymbol{M}(r,s)$ is the 6×6 matrix whose columns are the six fundamental solution vectors, and $\boldsymbol{C}(s)$ is a 6×1 column vector of arbitrary constants. For a homogeneous spherical layer, the matrix $\boldsymbol{M}$ was given in explicit form by Sabadini et al. (1982) and by Wu (1990). It can be partitioned with respect to the variables r and s into the form

$$\boldsymbol{M}(r,s) = \boldsymbol{M}^{(0)}(r) + \mu(s)\boldsymbol{M}^{(1)}(r) \; . \tag{5}$$

The explicit forms of the component matrices are

$$\boldsymbol{M}^{(0)} = \\ = \begin{pmatrix} \frac{nr^{n+1}}{2(2n+3)} & r^{n-1} & 0 & \frac{(n+1)r^{-n}}{2(2n-1)} & r^{-n-2} & 0 \\ \frac{(n+3)r^{n+1}}{2(2n+3)(n+1)} & \frac{r^{n-1}}{n} & 0 & \frac{-(n-2)r^{-n}}{2n(2n-1)} & \frac{-r^{-n-2}}{n+1} & 0 \\ \frac{n\varrho g r^{n+1}}{2(2n+3)} & \varrho g r^{n-1} & \varrho r^{n} & \frac{(n+1)\varrho g r^{-n}}{2(2n-1)} & \varrho g r^{-n-2} & \varrho r^{-n-1} \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & r^{n} & 0 & 0 & r^{-n-1} \\ \frac{4\pi G\varrho n r^{n+1}}{2(2n+3)} & 4\pi G\varrho r^{n-1} & (2n+1)r^{n-1} & \frac{4\pi G\varrho(n+1)r^{-n}}{2(2n-1)} & 4\pi G\varrho r^{-n-2} & 0 \end{pmatrix} , \tag{6}$$

$$\boldsymbol{M}^{(1)} = \begin{pmatrix} 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ \frac{(n^2-n-3)r^{n}}{2n+3} & 2(n-1)r^{n-2} & 0 & \frac{-(n^2+3n-1)r^{-n-1}}{2n-1} & -2(n+2)r^{-n-3} & 0 \\ \frac{n(n+2)r^{n}}{(2n+3)(n+1)} & \frac{2(n-1)r^{n-2}}{n} & 0 & \frac{(n-1)(n+1)r^{-n-1}}{n(2n-1)} & \frac{2(n+2)r^{-n-3}}{n+1} & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \end{pmatrix} , \tag{7}$$

where the appropriate expression for $g(r)$ must be substituted (see Appendix A). The three solutions associated with the last three columns of matrix $\boldsymbol{M}$ are singular at the origin and must be excluded for a homogeneous sphere (Wu & Peltier, 1982); however, for the homogeneous spherical layer considered here, they must be included.

In order to construct the propagator matrix for a multi-layered, incompressible viscoelastic sphere, the inverse of matrix $\boldsymbol{M}$ must be found. Following Spada et al. (1992), the inverse matrix $\boldsymbol{M}^{-1}$ may be partitioned into a form similar to that for $\boldsymbol{M}$:

$$\boldsymbol{M}^{-1}(r,s) = \boldsymbol{N}^{(0)}(r) + \frac{1}{\mu(s)}\boldsymbol{N}^{(-1)}(r) \; . \tag{8}$$

The explicit forms of the component matrices are

$$N^{(0)} =$$

$$= \frac{1}{2n+1} \begin{pmatrix} -2(n+1)(n+2)r^{-n-1} & 2n(n+1)(n+2)r^{-n-1} & 0 & 0 & 0 & 0 \\ \frac{n(n^2+3n-1)r^{-n+1}}{2n-1} & \frac{-n(n-1)(n+1)^2 r^{-n+1}}{2n-1} & 0 & 0 & 0 & 0 \\ -4\pi G\varrho r^{-n+1} & 0 & 0 & 0 & 0 & r^{-n+1} \\ 2n(n-1)r^n & 2n(n+1)(n-1)r^n & 0 & 0 & 0 & 0 \\ \frac{-(n+1)(n^2-n-3)r^{n+2}}{2n+3} & \frac{-n^2(n+1)(n+2)r^{n+2}}{2n+3} & 0 & 0 & 0 & 0 \\ 4\pi G\varrho r^{n+2} & 0 & 0 & 0 & (2n+1)r^{n+1} & -r^{n+2} \end{pmatrix} . \tag{9}$$

$$N^{(-1)} = \frac{n(n+1)}{2n+1} \begin{pmatrix} \frac{\varrho g r^{-n}}{n} & 0 & \frac{-r^{-n}}{n} & r^{-n} & \frac{\varrho r^{-n}}{n} & 0 \\ \frac{-\varrho g r^{-n+2}}{2(2n-1)} & 0 & \frac{r^{-n+2}}{2(2n-1)} & \frac{-(n-2)r^{-n+2}}{2(2n-1)} & \frac{-\varrho r^{-n+2}}{2(2n-1)} & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ \frac{\varrho g r^{n+1}}{n+1} & 0 & \frac{-r^{n+1}}{n+1} & -r^{n+1} & \frac{\varrho r^{n+1}}{n+1} & 0 \\ \frac{-\varrho g r^{n+3}}{2(2n+3)} & 0 & \frac{r^{n+3}}{2(2n+3)} & \frac{(n+3)r^{n+3}}{2(2n+3)} & \frac{-\varrho r^{n+3}}{2(2n+3)} & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \end{pmatrix} . \tag{10}$$

3. Layer propagator

Let the earth model be composed of N homogeneous incompressible spherical layers bounded by spheres of radii $r_1 = b < r_2 < ... < r_N = a$. In the kth ($k \geq 2$) layer, the relation between $\boldsymbol{Y}(r, s)$ at inner boundary r_{k-1} and outer boundary r_k can be written in the form (Gantmacher, 1959; Gilbert & Backus, 1966)

$$\boldsymbol{Y}(r_k, s) = \boldsymbol{P}(r_k, r_{k-1}, s)\boldsymbol{Y}(r_{k-1}, s) , \tag{11}$$

where the layer propagator matrix $\boldsymbol{P}(r_k, r_{k-1}, s)$ is given in terms of matrices $\boldsymbol{M}$ and $\boldsymbol{M}^{-1}$ as

$$\boldsymbol{P}(r_k, r_{k-1}, s) = \boldsymbol{M}(r_k, s)\boldsymbol{M}^{-1}(r_{k-1}, s) . \tag{12}$$

Making use of expressions (5) and (8) for $\boldsymbol{M}$ and $\boldsymbol{M}^{-1}$, respectively, $\boldsymbol{P}$ can be partitioned into

$$\boldsymbol{P}(r_k, r_{k-1}, s) = \boldsymbol{P}^{(0)}(r_k, r_{k-1}) + \mu_k(s)\boldsymbol{P}^{(1)}(r_k, r_{k-1}) + \frac{1}{\mu_k(s)}\boldsymbol{P}^{(-1)}(r_k, r_{k-1}) , \tag{13}$$

where the component matrices $\boldsymbol{P}^{(i)}(r_k, r_{k-1})$, $i = 0, \pm 1$, are expressed in terms of the component matrices of $\boldsymbol{M}$ and $\boldsymbol{M}^{-1}$ as

$$\begin{aligned} \boldsymbol{P}^{(0)}(r_k, r_{k-1}) &= \boldsymbol{M}^{(0)}(r_k)\boldsymbol{N}^{(0)}(r_{k-1}) + \boldsymbol{M}^{(1)}(r_k)\boldsymbol{N}^{(-1)}(r_{k-1}) , \\ \boldsymbol{P}^{(1)}(r_k, r_{k-1}) &= \boldsymbol{M}^{(1)}(r_k)\boldsymbol{N}^{(0)}(r_{k-1}) , \\ \boldsymbol{P}^{(-1)}(r_k, r_{k-1}) &= \boldsymbol{M}^{(0)}(r_k)\boldsymbol{N}^{(-1)}(r_{k-1}) . \end{aligned} \tag{14}$$

The non-zero elements of $\boldsymbol{P}^{(i)}(r_k, r_{k-1})$ are listed in Appendix A.

The solution in the central sphere ($0 \le r \le b$) is given by eqn.(4) which takes the particular form

$$\boldsymbol{Y}(r,s) = \boldsymbol{Z}(r,s)\boldsymbol{K}(s) \,, \tag{15}$$

where $\boldsymbol{Z}(r,s)$ is the 6×3 matrix whose columns are the first three columns of matrix $\boldsymbol{M}$, and $\boldsymbol{K}$ is a 3×1 column vector of constants. Note that, for a homogeneous inviscid core, matrix $\boldsymbol{Z}$ takes a specific form (e.g., Wu & Peltier, 1982). The solution in the kth spherical layer then follows from the upward continuation of the solution from the innermost sphere and the continuity of $\boldsymbol{Y}(r,s)$ at the interfaces at $r_1, \cdots, r_N$. This corresponds to the multiplication of the product of the layer propagator matrices with the inner sphere solution taken at $r = b$. In the kth layer, ($r_{k-1} \le r \le r_k$, $k \ge 2$), we obtain

$$\boldsymbol{Y}(r,s) = \boldsymbol{L}(r,b,s)\boldsymbol{Y}(b,s) \,, \tag{16}$$

where

$$\boldsymbol{L}(r,b,s) = \boldsymbol{P}(r,r_{k-1})\boldsymbol{P}(r_{k-1},r_{k-2}) \cdots \boldsymbol{P}(r_2,b) \,. \tag{17}$$

Substituting for $\boldsymbol{Y}(b,s)$ from eqn.(15) in eqn.(16), we have

$$\boldsymbol{Y}(r,s) = \boldsymbol{L}(r,b,s)\boldsymbol{Z}(b,s)\boldsymbol{K}(s) \,. \tag{18}$$

The constants in $\boldsymbol{K}(s)$ can be determined from the surface boundary conditions for an impulsive load (Longman, 1963; Farell, 1972):

$$\boldsymbol{B} = \begin{pmatrix} T_{rn}(a) \\ T_{\vartheta n}(a) \\ Q_n(a) \end{pmatrix} = \begin{pmatrix} \dfrac{-(2n+1)g(a)}{4\pi a^2} \\ 0 \\ \dfrac{-(2n+1)G}{a^2} \end{pmatrix} . \tag{19}$$

Using eqns.(15) and (19), eqn.(16) yields

$$\boldsymbol{B} = \boldsymbol{T}(s)\boldsymbol{K}(s) \,, \tag{20}$$

where $\boldsymbol{T}(s)$ is a 3×3 matrix defined by

$$\boldsymbol{T}(s) = \boldsymbol{L}(a,b,s)\boldsymbol{Z}(b,s) \tag{21}$$

with rows 1, 2 and 5 deleted. Determining $\boldsymbol{K}$ from eqn.(20) and substituting the corresponding expression into eqn.(18), gives the solution in the kth layer ($r_{k-1} \le r \le r_k$):

$$\boldsymbol{Y}(r,s) = \boldsymbol{L}(r,b,s)\boldsymbol{Z}(b,s)\boldsymbol{T}^{-1}(s)\boldsymbol{B} \,. \tag{22}$$

Denoting the matrix of cofactors of $\boldsymbol{T}(s)$ by $\boldsymbol{T}^{\dagger}(s)$, this solution can be written in an alternative form:

$$\boldsymbol{Y}(r,s) = \frac{\boldsymbol{W}(r,s)}{\det \boldsymbol{T}(s)} \,, \tag{23}$$

where

$$\boldsymbol{W}(r,s) = \boldsymbol{L}(r,b,s)\boldsymbol{Z}(b,s)\boldsymbol{T}^{\dagger}(s)\boldsymbol{B} \,. \tag{24}$$

4. Time-domain solution

According to Peltier (1985), Wolf (1985) and Wu(1990), the solution in the Laplace-transform domain can be expressed in the time domain as follows:

$$\boldsymbol{Y}(r,t) = \boldsymbol{Y}^E(r)\delta(t) + \sum_j \boldsymbol{Y}_j^V(r)e^{s_j t} \ . \tag{25}$$

where $\delta(t)$ is the delta function, $\boldsymbol{Y}^E(r)$ is the elastic amplitude:

$$\boldsymbol{Y}^E(r) = \lim_{s\to-\infty} \boldsymbol{Y}(r,s) \ . \tag{26}$$

$\boldsymbol{Y}_j^V(r)$ is the viscous amplitude spectrum and $-s_j$ the inverse relaxation time spectrum. The inverse relaxation times $-s_j$ can be determined from the roots of the secular determinant $\det \boldsymbol{T}(s) = 0$. The viscous amplitudes $\boldsymbol{Y}_j^V(r)$ are obtained using the residue theorem:

$$\boldsymbol{Y}_j^V(r) = \frac{\boldsymbol{W}(r,s_j)}{\frac{d}{ds}[\det \boldsymbol{T}(s)]\Big|_{s=s_j}} \ , \tag{27}$$

where simple roots s_j are implied.

5. The s derivative of secular determinant

The analytical form of the layer propagator matrix allows us to compute the s derivative of the secular determinant $\det \boldsymbol{T}(s)$ in eqn.(27) analytically. We first have

$$\frac{d}{ds}[\det \boldsymbol{T}(s)] = \sum_{i,j=1}^{3} \frac{\partial[\det \boldsymbol{T}(s)]}{\partial T_{ij}} \frac{dT_{ij}(s)}{ds} = \sum_{i,j=1}^{3} T_{ij}^{\dagger} \frac{dT_{ij}(s)}{ds} \ , \tag{28}$$

where $T_{ij}^{\dagger}$ is the cofactor of T_{ij}. The problem of finding an analytical s derivative of the secular determinant thus reduces to that of finding analytical s derivatives of the matrix elements T_{ij}. Taking into account eqn.(21), we can write

$$\frac{d\boldsymbol{T}(s)}{ds} = \frac{d\boldsymbol{L}(a,b,s)}{ds}\boldsymbol{Z}(b,s) + \boldsymbol{L}(a,b,s)\frac{d\boldsymbol{Z}(b,s)}{ds} \ . \tag{29}$$

The derivative in the first term on the right-hand side of eqn.(29) can be obtained by differentiating eqn.(17) with respect to s:

$$\begin{aligned}
\frac{d\boldsymbol{L}(a,b,s)}{ds} &= \frac{d\boldsymbol{P}(r_N,r_{N-1})}{ds}\boldsymbol{P}(r_{N-1},r_{N-2})\cdots\boldsymbol{P}(r_2,r_1) + \\
&+ \boldsymbol{P}(r_N,r_{N-1})\frac{\boldsymbol{P}(r_{N-1},r_{N-2})}{ds}\cdots\boldsymbol{P}(r_2,r_1) + \\
&+ \cdots \\
&+ \boldsymbol{P}(r_N,r_{N-1})P(r_{N-1},r_{N-2})\cdots\frac{d\boldsymbol{P}(r_2,r_1)}{ds} = \\
&= \sum_{k=2}^{N} \boldsymbol{L}(r_N,r_k,s)\frac{d\boldsymbol{P}(r_k,r_{k-1})}{ds}\boldsymbol{L}(r_{k-1},r_1,s) \ ,
\end{aligned} \tag{30}$$

where we consider that $\boldsymbol{L}(r_k, r_k, s)$ is equal to the 6×6 unit matrix. Differentiating also eqn.(13) with respect to s yields

$$\frac{d\boldsymbol{P}(r_k, r_{k-1}, s)}{ds} = \frac{d\mu_k(s)}{ds}\boldsymbol{P}^{(1)}(r_k, r_{k-1}) - \frac{1}{\mu_k^2(s)}\frac{d\mu_k(s)}{ds}\boldsymbol{P}^{(-1)}(r_k, r_{k-1}) \ , \tag{31}$$

where the s derivative of the shear modulus can be obtained from eqn.(3):

$$\frac{d\mu_k(s)}{ds} = \frac{\mu_k^2(s)}{\nu_k s^2} \ . \tag{32}$$

Equation (31) now takes the form

$$\frac{d\boldsymbol{P}(r_k, r_{k-1}, s)}{ds} = \frac{1}{\nu_k s^2}\left[\mu_k^2(s)\boldsymbol{P}^{(1)}(r_k, r_{k-1}) - \boldsymbol{P}^{(-1)}(r_k, r_{k-1})\right] \ . \tag{33}$$

The derivative in the second term on the right-hand side of eqn.(29) is similarly obtained by differentiating eqn.(5):

$$\frac{d\boldsymbol{Z}(b, s)}{ds} = \frac{\mu_1^2(s)}{\nu_1 s^2}\boldsymbol{Z}^{(1)}(b) \ , \tag{34}$$

where the 6×3 matrix $\boldsymbol{Z}^{(1)}(b)$ corresponds to matrix $\boldsymbol{M}^{(1)}(b)$ with columns 4, 5 and 6 deleted. Substituting eqns.(30), (33) and (34) into eqn.(29), we finally have

$$\frac{d\boldsymbol{T}(s)}{ds} = \frac{1}{s^2}\sum_{k=2}^{N}\left\{\boldsymbol{L}(r_N, r_k, s)\frac{1}{\nu_k}\left[\mu_k^2(s)\boldsymbol{P}^{(1)}(r_k, r_{k-1}) - \boldsymbol{P}^{(-1)}(r_k, r_{k-1})\right]\boldsymbol{L}(r_{k-1}, r_1, s)\right\}\boldsymbol{Z}(b, s) +$$

$$+ \frac{\mu_1^2(s)}{\nu_1 s^2}\boldsymbol{L}(a, b, s)\boldsymbol{Z}^{(1)}(b) \ . \tag{35}$$

Substituting the ij element of the last expression into eqn.(28) provides us the analytical form of the s derivative of the secular determinant.

6. Conclusion

This note was motivated by the question whether the layer propagator matrix for a self-gravitating incompressible linear-viscoelastic solid is sufficiently complicated that it can only be derived by a symbolic manipulation software. Our algebraic manipulations have shown that the layer propagator matrix $\boldsymbol{P}$ has a similar degree of complexity as, for instance, the matrix $\boldsymbol{M}$ of fundamental solutions or its inverse. The analytical form of the dependence of the layer propagator matrix on the Laplace-transform variable s has then allowed us to determine the s derivative of the secular determinant analytically. This is superior to a numerical determination of this derivative, particularly, for the cases where its poles and roots are located close to each other.

References

Amelung, F. & Wolf, D., 1994. Viscoelastic perturbations of the earth: significance of the incremental gravitational force in models of glacial isostasy, *Geophys. J. Int.*, **117**, 864–879.

$$
\begin{aligned}
P_{56}^{(0)} &= \frac{r_{k-1}}{2n+1}\left[h^n - h^{-n-1}\right] , \\
P_{61}^{(0)} &= 4\pi G\varrho_k\left[P_{11}^{(0)} - h^{n-1}\right] , \\
P_{62}^{(0)} &= 4\pi G\varrho_k P_{12}^{(0)} , \\
P_{66}^{(0)} &= h^{n-1} .
\end{aligned}
$$

The non-zero elements of the component matrix $P^{(-1)}(r_k, r_{k-1})$ are

$$
\begin{aligned}
P_{i1}^{(-1)} &= -\varrho_k g_{k-1} P_{i3}^{(-1)} , \qquad i = 1,2,3,6 \\
P_{13}^{(-1)} &= -\frac{n(n+1)r_{k-1}}{2(2n+1)(2n+3)}\left[h^{n+1} - h^{-n-2}\right] + \frac{n(n+1)r_{k-1}}{2(2n-1)(2n+1)}\left[h^{n-1} - h^{-n}\right] , \\
P_{23}^{(-1)} &= -\frac{r_{k-1}}{2(2n+1)(2n+3)}\left[(n+3)h^{n+1} + nh^{-n-2}\right] + \\
&+ \frac{r_{k-1}}{2(2n-1)(2n+1)}\left[(n+1)h^{n-1} + (n-2)h^{-n}\right] , \\
P_{33}^{(-1)} &= \varrho_k g_k P_{13}^{(-1)} , \\
P_{63}^{(-1)} &= 4\pi G\varrho_k P_{13}^{(-1)} , \\
P_{14}^{(-1)} &= \frac{n(n+1)r_{k-1}}{2(2n+1)(2n+3)}\left[nh^{n+1} + (n+3)h^{-n-2}\right] - \\
&- \frac{n(n+1)r_{k-1}}{2(2n-1)(2n+1)}\left[(n-2)h^{n-1} + (n+1)h^{-n}\right] , \\
P_{24}^{(-1)} &= \frac{n(n+3)r_{k-1}}{2(2n+1)(2n+3)}\left[h^{n+1} - h^{-n-2}\right] - \frac{(n+1)(n-2)r_{k-1}}{2(2n-1)(2n+1)}\left[h^{n-1} - h^{-n}\right] , \\
P_{34}^{(-1)} &= \varrho_k g_k P_{14}^{(-1)} , \\
P_{64}^{(-1)} &= 4\pi G\varrho_k P_{14}^{(-1)} , \\
P_{i5}^{(-1)} &= -\varrho_k P_{i3}^{(-1)} , \qquad i = 1,2,3,6 .
\end{aligned}
\tag{37}
$$

The non-zero elements of the component matrix $P^{(1)}(r_k, r_{k-1})$ are

$$
\begin{aligned}
P_{31}^{(1)} &= -\frac{2(n+1)(n+2)(n^2-n-3)}{(2n+1)(2n+3)r_{k-1}}\left[h^n - h^{-n-3}\right] + \frac{2n(n-1)(n^2+3n-1)}{(2n-1)(2n+1)r_{k-1}}\left[h^{n-2} - h^{-n-1}\right] , \\
P_{32}^{(1)} &= \frac{2n(n+1)(n+2)}{(2n+1)(2n+3)r_{k-1}}\left[(n^2-n-3)h^n + n(n+2)h^{-n-3}\right] - \\
&- \frac{2n(n-1)(n+1)}{(2n-1)(2n+1)r_{k-1}}\left[(n-1)(n+1)h^{n-2} + (n^2+3n-1)h^{-n-1}\right] , \\
P_{41}^{(1)} &= -\frac{2(n+2)}{(2n+1)(2n+3)r_{k-1}}\left[n(n+2)h^n + (n^2-n-3)h^{-n-3}\right] + \\
&+ \frac{2(n-1)}{(2n-1)(2n+1)r_{k-1}}\left[(n^2+3n-1)h^{n-2} + (n-1)(n+1)h^{-n-1}\right] , \\
P_{42}^{(1)} &= \frac{2n^2(n+2)^2}{(2n+1)(2n+3)r_{k-1}}\left[h^n - h^{-n-3}\right] - \frac{2(n-1)^2(n+1)^2}{(2n-1)(2n+1)r_{k-1}}\left[h^{n-2} - h^{-n-1}\right] .
\end{aligned}
\tag{38}
$$

The unperturbed gravitational acceleration within the kth layer ($r_{k-1} \leq r \leq r_k$, $k \geq 2$) is given by

$$g(r) = \frac{4\pi G}{3} \left[\varrho_k r + \frac{1}{r^2} \sum_{i=2}^{k} (\varrho_{i-1} - \varrho_i)\, r_{i-1}^3 \right] . \tag{39}$$

In the central sphere ($0 \leq r \leq r_1$), we have

$$g(r) = \frac{4\pi G}{3} \varrho_1 r . \tag{40}$$

```
c=================================================================
      program VEEN
c
c     Load induced relaxation of a multi-layered, incompressible,
c     Maxwell viscoelastic sphere
c
c     written by Zdenek Martinec (May 19, 1998)
c=================================================================
      implicit real*8 (a-h,o-z)
      dimension tmtrx(3,3),resi(3),resie(3)
      common /deg/ gfac,jdeg
      common /mdl/ rad(20),qden(20),qmu(20),qnu(20),grav(20)
      common /laplac/ s
      data qkappa /6.67d-11/
c
      gfac=16.d0*datan(1.d0)*qkappa
c
      open(68,file='veen.dat')
c----------------------
c     Model parameters
c----------------------
      open(55,file='modelh.dat')
      read(55,*) nlayer
      write(6,*) 'nlayer=',nlayer
      do 1 k=1,nlayer
      read(55,*) layer,rad(layer),qden(layer),qmu(layer),qnu(layer)
    1 write(6,*) layer,rad(layer),qden(layer),qmu(layer),qnu(layer)
      close(55)
c-------------------------------------------
c     Unperturbed gravitational acceleration
c     at the interfaces
c-------------------------------------------
      call GRAVITY(nlayer)
c
      write(6,*) 'Define jdeg:'
      read(5,*) jdeg
c
      ft=1000.d0*365.25d0*24.d0*3600.d0
c
      s1=-1.d-10
      s2=-1.d-11
c
      write(6,*) 'Define the number of s-steps:'
      read(5,*) ns
c----------------------------------------------------------------
      write(68,*)'Inverse relaxation times and viscous amplitudes'
c----------------------------------------------------------------
      hs=(s2-s1)/ns
      s=s1-hs
      do 50 is=1,ns+1
      s=s+hs
      call MAKET(nlayer,tmtrx)
      call DET33(tmtrx,det)
      write(6,*) 'det=',det
      trel=-1.d0/s/ft
c
      call DERDET(nlayer,ddet,resi)
      write(68,203) trel,s,det,(resi(i)/ddet,i=1,3)
  203 format(f8.5,2d14.5,2x,3d14.5)
   50 continue
      call AMPLH(nlayer,resi,resie)
      write(6,*)'Viscous and elastic amplitudes for homogeneous sphere:'
```



```
      write(6,*) resi
      write(6,*) resie
c-----------------------------------------
      write(68,*) 'Elastic amplitudes'
c-----------------------------------------
      ns=10
      s1=-1.d10
      s2=-1.d15
c
      hs=(s2-s1)/ns
      s=s1-hs
      do 51 is=1,ns+1
      s=s+hs
      call MAKET(nlayer,tmtrx)
      call DET33(tmtrx,det)
c
      call DERDET(nlayer,ddet,resi)
      write(68,204) s,det,(resi(i)/det,i=1,3)
  204 format(2d14.5,2x,3d14.5)
   51 continue
      stop
      end
c======================================
      subroutine MAKET(nlayer,tmtrx)
c
c     Matrix T
c======================================
      implicit real*8 (a-h,o-z)
      dimension tmtrx(3,3)
      dimension pl(6,6),p0(6,6),pp1(6,6),pm1(6,6)
      dimension em(6,3),em0(6,3),em1(6,3)
      dimension p(6,6),paux(6,6)
      dimension iarrt(3)
      data iarrt /3,4,6/
c
      do 1 i=1,6
      do 1 j=1,6
      p(i,j)=0.d0
    1 if(i.eq.j) p(i,i)=1.d0
c
      do 5 layer=2,nlayer
      call PROPAG(layer,pl,p0,pp1,pm1)
      do 3 i=1,6
      do 3 j=1,6
      sum=0.d0
      do 2 k=1,6
    2 sum=sum+pl(i,k)*p(k,j)
    3 paux(i,j)=sum
      do 4 i=1,6
      do 4 j=1,6
    4 p(i,j)=paux(i,j)
    5 continue
c
      call FUNDS(em,em0,em1)
      do 7 is=1,3
      i=iarrt(is)
      do 7 j=1,3
      sum=0.d0
      do 6 k=1,6
    6 sum=sum+p(i,k)*em(k,j)
    7 tmtrx(is,j)=sum
      return
```

```
      end
c=============================================
      subroutine DERDET(nlayer,ddet,resi)
c
c     Derivative of secular determinant det T
c     with respect to s
c=============================================
      implicit real*8 (a-h,o-z)
      dimension dery(3,3),dery1(6,6),dery2(6,3)
      dimension pl(6,6),p0(6,6),pp1(6,6),pm1(6,6)
      dimension em(6,6),em0(6,6),em1(6,6)
      dimension p(6,6),paux(6,6)
      dimension tmtrx(3,3),resi(*)
      dimension aux(3,3),aux1(3,3),tcross(3,3),bvec(3)
      dimension iarrt(3),iarrd(3)
      common /mdl/ rad(20),qden(20),qmu(20),qnu(20),grav(20)
      common /laplac/ s
      data iarrt /3,4,6/, iarrd /1,2,5/
c---------------------------------------------
c     Derivative of matrix L with respect to s
c---------------------------------------------
      do 1 i=1,6
      do 1 j=1,6
    1 dery1(i,j)=0.d0
c
      do 20 kk=2,nlayer
      do 2 i=1,6
      do 2 j=1,6
      p(i,j)=0.d0
    2 if(i.eq.j) p(i,i)=1.d0
c
      if(kk.eq.2) goto 7
      do 6 layer=2,kk-1
      call PROPAG(layer,pl,p0,pp1,pm1)
      do 4 i=1,6
      do 4 j=1,6
      sum=0.d0
      do 3 k=1,6
    3 sum=sum+pl(i,k)*p(k,j)
    4 paux(i,j)=sum
      do 5 i=1,6
      do 5 j=1,6
    5 p(i,j)=paux(i,j)
    6 continue
c
    7 qmk=qmu(kk)/(1.d0+qmu(kk)/qnu(kk)/s)
      call PROPAG(kk,pl,p0,pp1,pm1)
      do 8 i=1,6
      do 8 j=1,6
    8 pl(i,j)=(qmk*qmk*pp1(i,j)-pm1(i,j))/qnu(kk)/s/s
      do 10 i=1,6
      do 10 j=1,6
      sum=0.d0
      do 9 k=1,6
    9 sum=sum+pl(i,k)*p(k,j)
   10 paux(i,j)=sum
      do 11 i=1,6
      do 11 j=1,6
   11 p(i,j)=paux(i,j)
c
      if(kk.eq.nlayer) goto 16
      do 15 layer=kk+1,nlayer
```



```
      call PROPAG(layer,pl,p0,pp1,pm1)
      do 13 i=1,6
      do 13 j=1,6
      sum=0.d0
      do 12 k=1,6
   12 sum=sum+pl(i,k)*p(k,j)
   13 paux(i,j)=sum
      do 14 i=1,6
      do 14 j=1,6
   14 p(i,j)=paux(i,j)
   15 continue
c
   16 do 17 i=1,6
      do 17 j=1,6
   17 dery1(i,j)=dery1(i,j)+p(i,j)
   20 continue
c-------------------
c     Matrix L
c-------------------
      do 21 i=1,6
      do 21 j=1,6
      p(i,j)=0.d0
   21 if(i.eq.j) p(i,i)=1.d0
      do 25 layer=2,nlayer
      call PROPAG(layer,pl,p0,pp1,pm1)
      do 23 i=1,6
      do 23 j=1,6
      sum=0.d0
      do 22 k=1,6
   22 sum=sum+pl(i,k)*p(k,j)
   23 paux(i,j)=sum
      do 24 i=1,6
      do 24 j=1,6
   24 p(i,j)=paux(i,j)
   25 continue
c------------------------------------------------------------
c     Derivative of fundamental matrix Z with respect to s
c  ----------------------------------------------------------
      qm1=qmu(1)/(1.d0+qmu(1)/qnu(1)/s)
      call FUNDS(em,em0,em1)
      do 26 i=1,6
      do 26 j=1,3
   26 dery2(i,j)=qm1*qm1*em1(i,j)/qnu(1)/s/s
c---------------------------------------------- --
c     Derivative of matrix T with respect to s
c-------------------------------------------------
      do 28 is=1,3
      i=iarrt(is)
      do 28 j=1,3
      sum1=0.d0
      sum2=0.d0
      do 27 k=1,6
      sum1=sum1+dery1(i,k)*em(k,j)+p(i,k)*dery2(k,j)
   27 sum2=sum2+p(i,k)*em(k,j)
      dery(is,j)=sum1
   28 tmtrx(is,j)=sum2
c------------------------------------------ ---
c     derivative of det T with respect to s
c---- ----------------------------------------
      sum=0.d0
      do 29 i=1,3
      do 29 j=1,3
```

```
   29 sum=sum+COFACT(tmtrx,j,i)*dery(i,j)
      ddet=sum
c----------------------------------
c     vector of residui
c----------------------------------
      do 31 is=1,3
      i=iarrd(is)
      do 31 j=1,3
      sum=0.d0
      do 30 k=1,6
   30 sum=sum+p(i,k)*em(k,j)
   31 aux(is,j)=sum
c
      call ACROSS(tmtrx,tcross)
      do 33 i=1,3
      do 33 j=1,3
      sum=0.d0
      do 32 k=1,3
   32 sum=sum+aux(i,k)*tcross(k,j)
   33 aux1(i,j)=sum
c
      call BCL(nlayer,bvec)
      do 35 i=1,3
      sum=0.d0
      do 34 j=1,3
   34 sum=sum+aux1(i,j)*bvec(j)
   35 resi(i)=sum
      return
      end
c=================================================================
      subroutine FUNDS(em,em0,em1)
c
c     Matrix M of fundamental solutions that are regular
c     at the origin r=0 (M is evaluated at the top of a core.)
c=================================================================
      implicit real*8 (a-h,n-z)
      dimension em(6,3),em0(6,3),em1(6,3)
      common /deg/ gfac,jdeg
      common /mdl/ rad(20),qden(20),qmu(20),qnu(20),grav(20)
      common /laplac/ s
c
      r=rad(1)
      qd=qden(1)
      g0=grav(1)
      qm=qmu(1)/(1.d0+qmu(1)/qnu(1)/s)
      n=dfloat(jdeg)
c
      do 1 i=1,6
      do 1 j=1,3
      em0(i,j)=0.d0
    1 em1(i,j)=0.d0
c
      a111=n/2.d0/(n+n+3.d0)
      a121=(n+3.d0)/2.d0/(n+n+3.d0)/(n+1.d0)
      a131=qd*g0*n/2.d0/(n+n+3.d0)
      a141=n*(n+2.d0)/(n+n+3.d0)/(n+1.d0)
      a161=gfac*qd*n/2.d0/(n+n+3.d0)
      a122=1.d0/n
      a132=qd*g0
      a142=2.d0*(n-1.d0)/n
      a162=gfac*qd
      a133=qd
```



```
      a163=n+n+1.d0
      be1=(n*n-n-3.d0)/(n+n+3.d0)
      be2=2.d0*(n-1.d0)
c
ccc      pwrp=r**jdeg
      pwrp=1.d0
      em0(1,1)=a111*pwrp*r
      em0(2,1)=a121*pwrp*r
      em0(3,1)=a131*pwrp*r
      em0(6,1)=a161*pwrp*r
      em0(1,2)=pwrp/r
      em0(2,2)=a122*pwrp/r
      em0(3,2)=a132*pwrp/r
      em0(6,2)=a162*pwrp/r
      em0(3,3)=a133*pwrp
      em0(5,3)=pwrp
      em0(6,3)=a163*pwrp/r
c
      em1(3,1)=be1*pwrp
      em1(4,1)=a141*pwrp
      em1(3,2)=be2*pwrp/r/r
      em1(4,2)=a142*pwrp/r/r
c
      do 2 i=1,6
      do 2 j=1,3
    2 em(i,j)=em0(i,j)+qm*em1(i,j)
      return
      end
c=================================================================
      subroutine FUNDSI(em,em0,em1)
c
c     Matrix M of fundamental solutions that are regular
c     at the origin r=0 for a uniform inviscid core
c=================================================================
      implicit real*8 (a-h,o-z)
      dimension em(6,3),em0(6,3),em1(6,3)
      common /deg/ gfac,jdeg
      common /mdl/ rad(20),qden(20),qmu(20),qnu(20),grav(20)
c
      r=rad(1)
      qd=qden(1)
      g0=grav(1)
      dj=dfloat(jdeg)
c
      do 1 i=1,6
      do 1 j=1,3
      em0(i,j)=0.d0
    1 em1(i,j)=0.d0
c
      pwrp=r**jdeg
      em0(1,1)=-pwrp/g0
      em0(5,1)=pwrp
      em0(6,1)=2.d0*(dj-1.d0)*pwrp/r
      em0(2,2)=1.d0
      em0(1,3)=1.d0
      em0(3,3)=qd*g0
      em0(6,3)=gfac*qd
c
      do 2 i=1,6
      do 2 j=1,3
    2 em(i,j)=em0(i,j)
      return
```

```
      end
c=====================================================================
      subroutine PROPAG(layer,p,p0,pp1,pm1)
c
c     Layer propagator matrix
c
c     Inputs:
c         layer   ... index of the layer  (layer=1,2,..., nlayer)
c         jdeg    ... angular degree
c         rad     ... radii of the layer interfaces  (rad(1)<rad(2)<...)
c         qden    ... densities of layers
c         qmu     ... elastic shear moduli of layers
c         qnu     ... viscosities of layers
c         grav    ... gravitation at interfaces rad(.)
c         s       ... Laplace-transformed variable
c         gfac=4*pi*G
c
c     Outputs:
c         p           ... layer propagator matrix
c         p0,pm1,pm2  ... component matrices of the layer propagator
c=====================================================================
      implicit real*8 (a-h,n-z)
      dimension p(6,6),p0(6,6),pp1(6,6),pm1(6,6)
      common /deg/ gfac,jdeg
      common /mdl/ rad(20),qden(20),qmu(20),qnu(20),grav(20)
      common /laplac/ s
c
      r1=rad(layer-1)
      r2=rad(layer)
      qd=qden(layer)
      qm=qmu(layer)/(1.d0+qmu(layer)/qnu(layer)/s)
      g1=grav(layer-1)
      g2=grav(layer)
c
      h=r2/r1
      hn=h**jdeg
      hi=1.d0/hn/h
c
      n=dfloat(jdeg)
      u=1.d0
      d=2.d0
      t=3.d0
c
      do 1 i=1,6
      do 1 j=1,6
      p0(i,j)=0.d0
      pm1(i,j)=0.d0
    1 pp1(i,j)=0.d0
c---------------------------
c     Component matrix P{0}
c---------------------------
      p0(1,1)=-(n+u)*(n*(n+d)*hn*h+(n*n-n-t)*hi/h)/(n+n+u)/(n+n+t)
     *           +n*((n*n+t*n-u)*hn/h+(n*n-u)*hi*h)/(n+n-u)/(n+n+u)
      p0(1,2)= n*n*(n+u)*(n+d)*(hn*h-hi/h)/(n+n+u)/(n+n+t)
     *           -n*(n*n-u)*(n+u)*(hn/h-hi*h)/(n+n-u)/(n+n+u)
      p0(2,1)=-((n+d)*(n+t)*hn*h-(n*n-n-t)*hi/h)/(n+n+u)/(n+n+t)
     *           +((n*n+t*n-u)*hn/h-(n-u)*(n-d)*hi*h)/(n+n-u)/(n+n+u)
      p0(2,2)= n*(n+d)*((n+t)*hn*h+n*hi/h)/(n+n+u)/(n+n+t)
     *           -(n*n-u)*((n+u)*hn/h+(n-d)*hi*h)/(n+n-u)/(n+n+u)
      p0(3,1)=-n*(n+u)*(n+d)*qd*(g2*hn*h-g1*hi/h/h)/(n+n+u)/(n+n+t)
     *           +n*(n*n+t*n-u)*qd*(g2*hn/h-g1*hi)/(n+n-u)/(n+n+u)
     *           +(n+u)*(n*n-n-t)*qd*(g1*hn-g2*hi/h)/(n+n+u)/(n+n+t)
```



```
     *           -n*(n*n-u)*qd*(g1*hn/h/h-g2*hi*h)/(n+n-u)/(n+n+u)
     *           -gfac*qd*qd*r1*(hn-hi)/(n+n+u)
      p0(3,2)=qd*g2*p0(1,2)
      p0(3,3)=-(n+u)*((n*n-n-t)*hn+n*(n+d)*hi/h/h)/(n+n+u)/(n+n+t)
     *           +n*((n*n-u)*hn/h/h+(n*n+t*n-u)*hi)/(n+n-u)/(n+n+u)
      p0(3,4)=n*(n+u)*((n*n-n-t)*hn-(n+d)*(n+t)*hi/h/h)/(n+n+u)/(n+n+t)
     *    -n*(n+u)*((n-u)*(n-d)*hn/h/h-(n*n+t*n-u)*hi)/(n+n-u)/(n+n+u)
      p0(3,5)=-qd*(p0(3,3)-hi)
      p0(3,6)=qd*r1*(hn-hi)/(n+n+u)
      p0(4,3)=-n*(n+d)*(hn-hi/h/h)/(n+n+u)/(n+n+t)
     *           +(n*n-u)*(hn/h/h-hi)/(n+n-u)/(n+n+u)
      p0(4,4)= n*(n+d)*(n*hn+(n+t)*hi/h/h)/(n+n+u)/(n+n+t)
     *           -(n*n-u)*((n-d)*hn/h/h+(n+u)*hi)/(n+n-u)/(n+n+u)
      p0(4,1)=-qd*g1*p0(4,3)
      p0(4,5)=-qd*p0(4,3)
      p0(5,1)=-gfac*qd*r1*(hn-hi)/(n+n+u)
      p0(5,5)=hi
      p0(5,6)=r1*(hn-hi)/(n+n+u)
      p0(6,1)=gfac*qd*(p0(1,1)-hn/h)
      p0(6,2)=gfac*qd*p0(1,2)
      p0(6,6)=hn/h
c--  --------------------------
c     Component matrix P{-1}
c------------------------ ----
      pm1(1,3)=-n*(n+u)*r1*(hn*h-hi/h)/d/(n+n+u)/(n+n+t)
     *            +n*(n+u)*r1*(hn/h-hi*h)/d/(n+n-u)/(n+n+u)
      pm1(2,3)=-r1*((n+t)*hn*h+n*hi/h)/d/(n+n+u)/(n+n+t)
     *            +r1*((n+u)*hn/h+(n-d)*hi*h)/d/(n+n-u)/(n+n+u)
      pm1(3,3)=qd*g2*pm1(1,3)
      pm1(6,3)=gfac*qd*pm1(1,3)
      pm1(1,1)=-qd*g1*pm1(1,3)
      pm1(2,1)=-qd*g1*pm1(2,3)
      pm1(3,1)=-qd*g1*pm1(3,3)
      pm1(6,1)=-qd*g1*pm1(6,3)
      pm1(1,4)=n*(n+u)*r1*(n*hn*h+(n+t)*hi/h)/d/(n+n+u)/(n+n+t)
     *           -n*(n+u)*r1*((n-d)*hn/h+(n+u)*hi*h)/d/(n+n-u)/(n+n+u)
      pm1(2,4)=n*(n+t)*r1*(hn*h-hi/h)/d/(n+n+u)/(n+n+t)
     *           -(n+u)*(n-d)*r1*(hn/h-hi*h)/d/(n+n-u)/(n+n+u)
      pm1(3,4)=qd*g2*pm1(1,4)
      pm1(6,4)=gfac*qd*pm1(1,4)
      pm1(1,5)=-qd*pm1(1,3)
      pm1(2,5)=-qd*pm1(2,3)
      pm1(3,5)=-qd*pm1(3,3)
      pm1(6,5)=-qd*pm1(6,3)
c------------------------------
c     Component matrix P{+1}
c------------------------------
      pp1(3,1)=-d*(n+u)*(n+d)*(n*n-n-t)*(hn-hi/h/h)/r1/(n+n+u)/(n+n+t)
     *            +d*n*(n-u)*(n*n+t*n-u)*(hn/h/h-hi)/r1/(n+n-u)/(n+n+u)
      pp1(3,2)=
     * d*n*(n+u)*(n+d)*((n*n-n-t)*hn+n*(n+d)*hi/h/h)/r1/(n+n+u)/(n+n+t)
     *-d*n*(n*n-u)*((n*n-u)*hn/h/h+(n*n+t*n-u)*hi)/r1/(n+n-u)/(n+n+u)
      pp1(4,1)=-d*(n+d)*(n*(n+d)*hn+(n*n-n-t)*hi/h/h)/r1/(n+n+u)/(n+n+t)
     *           +d*(n-u)*((n*n+t*n-u)*hn/h/h+(n*n-u)*hi)/r1/(n+n-u)/(n+n+u)
      pp1(4,2)=d*n*n*(n+d)*(n+d)*(hn-hi/h/h)/r1/(n+n+u)/(n+n+t)
     *           -d*(n*n-u)*(n*n-u)*(hn/h/h-hi)/r1/(n+n-u)/(n+n+u)
c-----    --  - ----------------
c     Layer propagator matrix P
c-  ---------- -- -------------
      do 2 i=1,6
      do 2 j=1,6
    2 p(i,j)=p0(i,j)+qm*pp1(i,j)+pm1(i,j)/qm
```

Zentralbibliothek 000977245
GFZ Potsdam B 103